

Лабораторная работа №3

Моделирование стохастических процессов

Чемоданова Ангелина Александровна

19 февраля 2025

Российский университет дружбы народов имени Патриса Лумумбы, Москва, Россия

Информация

- Чемоданова Ангелина Александровна
- Студентка НФИбд02-22
- Российский университет дружбы народов имени Патриса Лумумбы
- 1132226443@pfur.ru
- <https://github.com/aachemodanova>



Смоделировать систему массового обслуживания и построить график поведения длины очереди.

1. Реализовать модель системы массового обслуживания $M/M/1$ и рассчитать характеристики модели.
2. Построить график поведения длины очереди.

$M|M|1$ — однолинейная СМО с накопителем бесконечной ёмкости. Поступающий поток заявок — пуассоновский с интенсивностью λ . Времена обслуживания заявок — независимые в совокупности случайные величины, распределённые по экспоненциальному закону с параметром μ . Рассмотрим реализацию данной модели с параметрами системы: $\lambda = 30.0$ и $\mu = 33.0$.

Реализация модели М/М/1

```
# создание объекта Simulator
set ns [new Simulator]

# открытие на запись файла out.tr для регистрации событий
set tf [open out.tr w]
$ns trace-all $tf

# задаём значения параметров системы
set lambda 30.0
set mu 33.0

# размер очереди для M|M|1 (для M|M|1|R: set qsize R)
set qsize 100000

# устанавливаем длительность эксперимента
set duration 1000.0

# задаём узлы и соединяем их симплексным соединением
# с полосой пропускания 100 Кб/с и задержкой 0 мс,
# очереди с обслуживанием типа DropTail
set n1 [$ns node]
set n2 [$ns node]

set link [$ns simplex-link $n1 $n2 100kb 0ms DropTail]

# наложение ограничения на размер очереди:
$ns queue-limit $n1 $n2 $qsize

# задаём распределения интервалов времени
# поступления пакетов и размера пакетов
set InterArrivalTime [new RandomVariable/Exponential]
$InterArrivalTime set avg_ [expr 1/$lambda]
set pktSize [new RandomVariable/Exponential]
$pktSize set avg_ [expr 100000.0/(8*$mu)]

# задаём агент UDP и присоединяем его к источнику,
# задаём размер пакета
set src [new Agent/UDP]
$src set packetSize 100000
$ns attach-agent $n1 $src

# задаём агент-приёмник и присоединяем его
set sink [new Agent/Null]
$ns attach-agent $n2 $sink
$ns connect $src $sink

# мониторинг очереди
set qmon [$ns monitor-queue $n1 $n2 [open qn.out w] 0.1]
$link queue-sample-timeout
```

Рис. 1: Реализация модели М/М/1

```
# процедура finish закрывает файлы трассировки
proc finish {} {
    global ns tf
    $ns flush-trace
    close $tf
    exit 0
}

# процедура случайного генерирования пакетов
proc sendpacket {} {
    global ns src InterArrivalTime pktSize
    set time [$ns now]
    $ns at [expr $time + [$InterArrivalTime value]] "sendpacket"
    set bytes [expr round ([$pktSize value])]
    $src send $bytes
}

# планировщик событий
$ns at 0.0001 "sendpacket"
$ns at $duration "finish"

# расчет загрузки системы и вероятности потери пакетов
set rho [expr $lambda/$mu]
set ploss [expr (1-$rho)*pow($rho,$qsize)/(1-pow($rho,($qsize+1)))]
puts "Теоретическая вероятность потери = $ploss"

set aveq [expr $rho*$rho/(1-$rho)]
puts "Теоретическая средняя длина очереди = $aveq"
# запуск модели
$ns run
```

Рис. 2: Реализация модели М/М/1


После запуска модели, были получены данные характеристики системы.

```
openmodelica@openmodelica-VirtualBox:~/mip/lab-ns$ ns lab3_mm1.tcl  
Теоретическая вероятность потери = 0.0  
Теоретическая средняя длина очереди = 9.0909090909090864
```

Рис. 3: Полученные характеристики системы

Построить график поведения длины очереди

В каталоге с проектом создадим отдельный файл, например, graph_plot. Откроем его на редактирование и добавим следующий код, обращая внимание на синтаксис GNUplot:



```
#!/usr/bin/gnuplot -persist
# задаём текстовую кодировку,
# тип терминала, тип и размер шрифта

set encoding utf8
set term pdfcairo font "Arial,9"

# задаём выходной файл графика
set out 'qm.pdf'

# задаём название графика
set title "График средней длины очереди"

# задаём стиль линии
set style line 2

# подписи осей графика
set xlabel "t"
set ylabel "Пакеты"

# построение графика, используя значения
# 1-го и 5-го столбцов файла qm.out
plot "qm.out" using ($1):($5) with lines lt rgb "yellow" title "Размер очереди (в пакетах)", \
      "qm.out" using ($1):($5) smooth csplines title " Приближение сплайном ", \
      "qm.out" using ($1):($5) smooth bezier lt rgb "red" title " Приближение Безье "
```

Рис. 4: Скрипт для построения графика поведения длины очереди

Построить график поведения длины очереди

Сделаем файл исполняемым. После компиляции файла с проектом, запустим скрипт в созданном файле graph_plot, который создаст файл qm.pdf с результатами моделирования. Изменение прав доступа файла graph_plot. И запуск скрипта.

```
openmodelica@openmodelica-VirtualBox:~/mip/lab-ns$ chmod +x graph_plot  
openmodelica@openmodelica-VirtualBox:~/mip/lab-ns$ ./graph_plot
```

Рис. 5: Изменение прав доступа файла graph_plot. И запуск скрипта

График поведения длины очереди

Результаты моделирования. График поведения длины очереди.




Рис. 6: График поведения длины очереди

Мы смоделировали систему массового обслуживания и построили график поведения длины очереди.